

DUM č. 16 v sadě

28. Inf-4 Jednoduchá hra Had ve Flashi (ActionScript)

Autor: Robert Havlásek

Datum: 07.06.2013

Ročník: 5AV

Anotace DUMu: Flash - příklad: Zavedení proměnných pro počítání životů a sněžených jablek. Jejich inkrementace / dekrementace ve vhodných místech programu. Bonusový život za jistý počet jablek.\n

Materiály jsou určeny pro bezplatné používání pro potřeby výuky a vzdělávání na všech typech škol a školských zařízení. Jakékoliv další využití podléhá autorskému zákonu.



INVESTICE DO ROZVOJE VZDĚLÁVÁNÍ

V tomto DUMu zavedeme proměnnou pro počítání životů hada. Rovněž budeme počítat sněžená jablka. Zavedeme pravidlo, že při jistém počtu jablek dostaneme (jako bonus) jeden nový život.

Počítání životů, „univerzální“ smrt

Při spuštění programu (nejlépe v obsluze události onClipEvent(load)) je nutné zavést u hlavy hada proměnnou zivoty. *Pozn: Na proměnnou zivoty lze nahlížet i jako na vlastnost objektu _root.hlava. Lze na ni nahlížet i jako na index asociativního pole.*

Celý systém počítání životů je zatím interní záležitostí; životy se nikde nevypisují (to až v příštím a přespříštím DUMu) a nic neovlivňují (GAME OVER bude až v posledním DUMu).

Pedagogická poznámka: Určení defaultního počtu životů, které had na začátku dostane, nechám obvykle na studentech.

Tedy:

```
onClipEvent(load) {
    var telox:Array = Array();
    var teloy:Array = Array();
    var defaultnidelka:Number = 4;
    var zivoty:Number = 5;
    ...
}
```

Umřeme-li, sníží se nám počet životů o jeden. Tedy v kódu plochy ve funkci smrt() připišeme `_root.hlava.zivoty--`. Při této příležitosti by bylo vhodné udělat funkci `smrt()` „univerzální“, tedy takovou, aby fungovala na všechny objekty, jež dostane ve svém parametru. Tedy původní

```
function smrt() {
    _root.hlava._x=160;
    _root.hlava._y=100;
    _root.hlava.smer=4;
    for (var m=0;m<_root.hlava.telox.length;m++)
        {_root.hlava.telox[m]=_root.hlava._x;
         _root.hlava.teloy[m]=_root.hlava._y;
        }
}
```

bychom upravili na

```
function smrt(koho:Object) {
    koho.zivoty--;
    koho._x=160;
    koho._y=100;
    koho.smer=4;
    for (var m=0;m<koho.telox.length;m++)
        {koho.telox[m]=koho._x;
         koho.teloy[m]=koho._y;
        }
}
```

Souhrnem: Zavedli jsme parametr `koho`, texty `_root.hlava` jsme nahradili za `koho`.

Pozor, změnu je třeba aplikovat i v kódu hlavy při volání. Namísto

```
if (obsazeno==1) _root.smrt();
```

je nutné psát

```
if (obsazeno==1) _root.smrt(_root.hlava);
```

nebo lépe

```
if (obsazeno==1) _root.smrt(this);
```

Počítání sněžených jablek

Při spuštění programu (nejlépe v obsluze události `onClipEvent(load)`) u hlavy hada zavedeme i proměnnou `jablek`. Zavedeme též proměnnou, která bude udávat, při kolika sněžených jablkách se zvýší počet životů o jeden, pojmenujme ji `pocetjablekprobonus`:

Tedy:

```
onClipEvent(load) {
    var telox:Array = Array();
    var teloy:Array = Array();
    var defaultnidelka:Number = 4;
    var zivoty:Number = 5;
    var jablek:Number = 0;
    var pocetjablekprobonus:Number = 5;
    ...
}
```

Pedagogická poznámka: Počet sněžených jablek v daném levelu hry bychom mohli celkem bez problémů určit z délky hada. Jenže při přechodu do vyššího levelu (viz poslední DUM) se bude délka hada nastavovat na defaultní délku, ale počet sněžených jablek se resetovat nesmí – chceme, aby se bonus počítal napříč levely...

Při sněžení jablka připišeme inkrementaci proměnné `jablek`:

```
// KONTROLA, ZDA PRAVE NEJIME JABLKO:
if ((_root.hlava._x==_root.jablko1._x)&&(_root.hlava._y==_root.jablko1._y))
    {jablek++;
    _root.tvor_jablko("jablko1",41);
    var p=_root.hlava.telox.length;
    _root.hlava.telox.push(_root.hlava.telox[p]);
    _root.hlava.teloy.push(_root.hlava.teloy[p]);
    }
```

Zapišeme poté i podmínku, která v případě, že počet sněžených jablek je zrovna dělitelný počtem, za něj získáme bonusový život, přidá onen život:

```
// KONTROLA, ZDA PRAVE NEJIME JABLKO:
if ((_root.hlava._x==_root.jablko1._x)&&(_root.hlava._y==_root.jablko1._y))
    {jablek++;
    if ((jablek % pocetjablekprobonus) == 0) zivoty++;
    _root.tvor_jablko("jablko1",41);
    var p=_root.hlava.telox.length;
    _root.hlava.telox.push(_root.hlava.telox[p]);
    _root.hlava.teloy.push(_root.hlava.teloy[p]);
    }
```

Používáme přitom operátor `%`, který znamená zbytek po dělení.

Vzhledem k tomu, že nemáme žádnou kontrolu funkčnosti kódu, doporučuji při zisku bonusového života vypsát `trace("bonus")`, tedy:

```
// KONTROLA, ZDA PRAVE NEJIME JABLKO:
if ((_root.hlava._x==_root.jablko1._x)&&(_root.hlava._y==_root.jablko1._y))
    {jablek++;
    if ((jablek % pocetjablekprobonus) == 0) {zivoty++;trace("bonus");}
    _root.tvor_jablko("jablko1",41);
    var p=_root.hlava.telox.length;
    _root.hlava.telox.push(_root.hlava.telox[p]);
    _root.hlava.teloy.push(_root.hlava.teloy[p]);
    }
```

Pro přehlednost celý kód plochy:

```
var zdix:Array = Array();
var zdiy:Array = Array();
var klonzdi:Array = Array();
// TVORBA POLE ZDI OKOLO:
for (i=0; i<=310; i=i+10) {zdix.push(i);zdiy.push(0);} // horni rada
for (i=0; i<=310; i=i+10) {zdix.push(i);zdiy.push(190);} // spodni rada
for (i=0; i<=190; i=i+10) {zdix.push(0);zdiy.push(i);} // leva rada
for (i=0; i<=190; i=i+10) {zdix.push(310);zdiy.push(i);} // prava rada
// FYZICKE KLONOVANI ZDI OKOLO:
for (k=0; k<zdix.length; k++){
    _root.vzorzdi._x=zdix[k];
    _root.vzorzdi._y=zdiy[k];
    duplicateMovieClip(_root.vzorzdi,"klonzdi["+k+"]",50+getNextHighestDepth());}
// NAHODNE ZDI A JEJICH KONTROLA:
for (k=0; k<20; k=k+1){
    _root.vzorzdi._x=10+10*random(30);
    _root.vzorzdi._y=10+10*random(18);
    // KONTROLA OBSAZENOSTI:
    var obsazeno=0;
    for (m=0; m<zdix.length; m=m+1)
        if ((_root.vzorzdi._x==zdix[m])&&(_root.vzorzdi._y==zdiy[m]))
            {obsazeno=1;break;}
    if ( (obsazeno==1) or
        ((_root.vzorzdi._x==160)&&(_root.vzorzdi._y==100)) ) {k=k-1}
    else {zdix.push(_root.vzorzdi._x);
        zdiy.push(_root.vzorzdi._y);
        duplicateMovieClip(_root.vzorzdi,"klonzdi["+zdix.length-1+"]",getNextHighestDepth());}
    }

tvor_jablko("jablko1",41);

function tvor_jablko(jmeno:String, hloubka:Number):Number {
    // TVORBA JABLKA:
    this.createEmptyMovieClip(jmeno, hloubka);
    var j=this.getInstanceAtDepth(hloubka);
        j.beginFill(0xFF0000);
    drawCircle(j, 5, 5, 5);
    j.endFill();
    do {
        j._x=10+10*random(30);
        j._y=10+10*random(18);
        var obsazeno=0;
        for (m=0; m<zdix.length; m=m+1)
            if ((j._x==zdix[m])&&(j._y==zdiy[m]))
                {obsazeno=1;break;}
        for (m=0; m<_root.hlava.telox.length; m=m+1)
            if ((j._x==_root.hlava.telox[m])&&(j._y==_root.hlava.teloy[m]))
                {obsazeno=1;break;}
        } while (obsazeno==1);
    return 0;
} // konec funkce tvor_jablko

function drawCircle(mc:MovieClip, x:Number, y:Number, r:Number):Void {
    mc.moveTo(x+r, y);
    mc.curveTo(r+x, Math.tan(Math.PI/8)*r+y, Math.sin(Math.PI/4)*r+x,
Math.sin(Math.PI/4)*r+y);
    mc.curveTo(Math.tan(Math.PI/8)*r+x, r+y, x, r+y);
    mc.curveTo(-Math.tan(Math.PI/8)*r+x, r+y, -Math.sin(Math.PI/4)*r+x,
Math.sin(Math.PI/4)*r+y);
    mc.curveTo(-r+x, Math.tan(Math.PI/8)*r+y, -r+x, y);
```

v ploše:

```
mc.curveTo(-r+x, -Math.tan(Math.PI/8)*r+y, -Math.sin(Math.PI/4)*r+x,
-Math.sin(Math.PI/4)*r+y);
mc.curveTo(-Math.tan(Math.PI/8)*r+x, -r+y, x, -r+y);
mc.curveTo(Math.tan(Math.PI/8)*r+x, -r+y, Math.sin(Math.PI/4)*r+x, -
Math.sin(Math.PI/4)*r+y);
mc.curveTo(r+x, -Math.tan(Math.PI/8)*r+y, r+x, y);
} // konec funkce drawCircle

function smrt(koho:Object) {
    koho.zivoty--;
    koho._x=160;
    koho._y=100;
    koho.smer=4;
    for (var m=0;m<koho.telox.length;m++)
        {koho.telox[m]=koho._x;
         koho.teloy[m]=koho._y;
        }
}
```

A i celý kód hlavy:

```
on (keyPress "<Left>") {smer=2;}
on (keyPress "<Right>") {smer=0;}
on (keyPress "<Up>") {smer=1;}
on (keyPress "<Down>") {smer=3;}

onClipEvent (enterFrame)
{ if (smer==0) {_root.hlava._x = _root.hlava._x + 10;}
  if (smer==1) {_root.hlava._y = _root.hlava._y - 10;}
  if (smer==2) {_root.hlava._x = _root.hlava._x - 10;}
  if (smer==3) {_root.hlava._y = _root.hlava._y + 10;}
  if (_root.hlava._x>=320) {_root.hlava._x=0;}
  if (_root.hlava._y>=200) {_root.hlava._y=0;}
  if (_root.hlava._x<0) {_root.hlava._x=310;}
  if (_root.hlava._y<0) {_root.hlava._y=190;}
  // KONTROLA, ZDA PRAVE NEJIME JABLKO:
  if ((_root.hlava._x==_root.jablkol._x)&&(_root.hlava._y==_root.jabl
kol._y))
    {jablek++;
     if ((jablek % pocetjablekprobonus) == 0) {zivoty++;trace("bonus");}
     _root.tvor_jablko("jablko1",41);
     var p=_root.hlava.telox.length;
     _root.hlava.telox.push(_root.hlava.telox[p]);
     _root.hlava.teloy.push(_root.hlava.teloy[p]);
    }
  // KONTROLA NARAZENI HLAVY DO ZDI:
  var obsazeno=0;
  for (m=0; m<_root.zdix.length; m=m+1)
    if ((_root.hlava._x==_root.zdix[m])&&(_root.hlava._y==_root.zdiy[m]))
      {obsazeno=1;break;}
  // A I KONTROLA NARAZENI DO SVEHO TELA (JEN POKUD SE HYBEME):
  if (smer<4) for (m=0; m<_root.hlava.telox.length; m=m+1)
    if ((_root.hlava._x==_root.hlava.telox[m])&&(_root.hlava._y==_root.hla
va.teloy[m]))
      {obsazeno=1;break;}
  if (obsazeno==1) _root.smrt(this);
/*   if ((_root.hlava._x>=320) || (_root.hlava._y>=200) ||
      (_root.hlava._x<0) || (_root.hlava._y<0)) _root.smrt();*/
  // VYPOCET NOVEHO TELA:
  for (m=telox.length-2;m>=0;m--) {telox[m+1]=telox[m];
                                   teloy[m+1]=teloy[m];}

  telox[0]=_root.hlava._x;
  teloy[0]=_root.hlava._y;
```

v objektu hlava:

```
// KRESLENI TELA:
for (var m=0;m<telox.length;m++)
    {removeMovieClip("klontelo["+m+"]");
      _root.telovzor._x=telox[m];
      _root.telovzor._y=teloy[m];
      duplicateMovieClip(_root.telovzor,"klontelo["+m+"]",m)
    }
} // konec onClipEvent(enterFrame)

onClipEvent(load) {
  var telox:Array = Array();
  var teloy:Array = Array();
  var defaultnidelka:Number = 4;
  var zivoty:Number = 5;
  var jablek:Number = 0;
  var pocetjablekprobonus:Number = 5;
  _root.telovzor._x=_root.hlava._x; //posuneme vzor do pozice hlavy
  _root.telovzor._y=_root.hlava._y;
  for (var i=0; i<defaultnidelka; i++) //zduplikujeme (defaultnidelka)-krat
    {telox.push(_root.hlava._x);
      teloy.push(_root.hlava._y);
      duplicateMovieClip(_root.telovzor,"klontelo"+i+"",i)
    } //konec cyklu for
} // konec onClipEvent(load)
```